
Simulação de Monte Carlo 
     
   
Este assunto ficou vazio durante certo período.  Foi difícil eu encontrar um bom exemplo para ele, pois a Simulação de Monte 
Carlo é um campo muito amplo.  Que exemplo seria apropriado aqui?  Não é uma questão fácil. Entretanto, devido à grande 
demanda sobre este tópico, decidi propor uma versão "Mickey Mouse" da Simulação de Monte Carlo.  Por gentileza note que este 
exemplo faz um monte de hipóteses livres que podem ou não refletir o cenário do mundo real.   

O que é uma Simulação de Monte Carlo?  Bem, pense sobre ela como um processo computacional que utiliza números aleatórios 
para produzir resultado(s).  Assim em vez de ter entradas fixas, distribuições de probabilidade são atribuídas a algumas ou todas 
as entradas.  Isto gerará uma distribuição de probabilidade para a saída após a simulação ser executada.   

Aqui está um exemplo.  Uma empresa que vende produto X sob um mercado1 puro/perfeitamente competitivo quer saber a 
distribuição de probabilidade dos lucros deste produto e a probabilidade que a empresa perca dinheiro quando comercializá-lo.   

A equação para o lucro é:     LT = RT - CT = (Q*PU) - (Q*CV+CF)   

Hipóteses:   

• A Quantidade Demandada (Q) flutua entre 8.000 e 12.000 unidades e está distribuída uniformemente.   
• O Custo Variável (CV) está normalmente distribuído (com média = 7, Desv = 2) truncado em ambos os lados (com um 

mínimo de 7 / 2 e um máximo de 10).   
• O Preço de Mercado (P) está normalmente distribuído (com média = 10, Desv = 3) truncado no lado esquerdo (com um 

mínimo de 1).   
• O Custo Fixo (CF) é $5.000.  

                                                                                                                  
   
Resultado:   
  
O lucro médio para este investimento é $29,546 como mostrado nas células G25 após 50.000 iterações ser executadas.  A 
probabilidade que o lucro do investimento rejeite de ser negativo (perder dinheiro) é 22,28% como mostrado na célula  C24. A 
distribuição de probabilidade do lucro > X está mostrada na coluna F e G.  Por exemplo, há 65% de chance que o lucro seja maior 
que $12.481. A distribuição de probabilidade é bem normal como mostrado na figura.  A média está também muito próxima da 
mediana. Isto é devido a distribuição de probabilidade que atribuímos às variáveis. 
 
 
 

  

 

 

 

 

 

 

 

                                                                  
1 Sob um mercado de perfeita competitividade, a empresa não tem a influência a afetar o preço deste produto - a empresa 
consegue o preço de mercado como aquele dado, dP/dQ = 0.   



Figura 1  

 

 

 

 

 

 

 

 

 

 



Figura 2  

 

 
 

Option Explicit 
Option Base 1 
 
'********************************************************************************** 
'*                           Executa Simulação de Monte Carlo                     * 
'********************************************************************************** 
Sub MonteCarlo() 
 
    Dim Iteracao As Long, i As Long 
    Dim Q As Double, P As Double, RT As Double 
    Dim CV As Double, CF As Double, CT As Double 
    Dim DesvPadCV As Double, CVMedio As Double, DesvPadLucro As Double, LucroMedio As 
Double 
    Dim MinQ As Double, MaxQ As Double, LucroTotalMedio As Double, SomaDoLucroTotal As 
Double 
    Dim LucroX As Double, NumeroDeContagem As Double 
    Iteracao = Range("C3").Value 
    CF = Range("C7").Value 
    MinQ = Range("C13").Value 
    MaxQ = Range("C14").Value 
    CVMedio = Range("C15").Value 
    DesvPadCV = Range("C16").Value 
    LucroMedio = Range("C17").Value 
    DesvPadLucro = Range("C18").Value 
    LucroX = Range("B24").Value 
     
    ReDim LT(Iteracao) As Double 
     
    SomaDoLucroTotal = 0 
    NumeroDeContagem = 0 
    For i = 1 To Iteracao:        Cells(12, 3) = i 
        CV = Truncate_Normal_CV(CVMedio, DesvPadCV, CVMedio / 2, LucroMedio) 
        P = Truncate_Normal_P(LucroMedio, DesvPadLucro, 1) 
        Q = Int((MaxQ - MinQ + 1) * Rnd + MinQ) 
        CT = CF + CV * Q 
        RT = P * Q 



        LT(i) = RT - CT 
         
        'Assinalar que o seguinte fará a simulação rodar mais rápido 
        Cells(5, 3) = Q 
        Cells(6, 3) = P 
        Cells(8, 3) = CV 
        Cells(9, 3) = CT 
        Cells(10, 3) = RT 
        Cells(11, 3) = LT(i) 
         
         
        If LT(i) > LucroX Then NumeroDeContagem = NumeroDeContagem + 1 
        SomaDoLucroTotal = SomaDoLucroTotal + LT(i) 
    Next i 
    LucroTotalMedio = SomaDoLucroTotal / Iteracao 
    Cells(25, 7) = LucroTotalMedio 
    Cells(24, 3) = 1 - NumeroDeContagem / Iteracao 
    Call Sort(Iteracao, LT) 
    Call Hist(Iteracao, 40, LT(1), LT(Iteracao), LT) 
    For i = 1 To 20 
        Cells(i + 3, 6) = 1 - (0.05 * i) 
        Cells(i + 3, 7) = LT(Int(Iteracao / 20 * i)) 
    Next i 
        Cells(3, 6) = "Perto de 100%" 
        Cells(13, 6) = "Mediana = 50%" 
        Cells(23, 6) = "Perto de 0%" 
        Cells(3, 7) = LT(1) 
End Sub 
 
'********************************************************************************** 
'*   Retorna números aleatórios de uma Distribuição Normal Truncada para    CV    * 
'********************************************************************************** 
Function Truncate_Normal_CV(XMedio, DesvPadX, LimiteEsquerdo, LimiteDireito) 
    Dim x As Double 
    Dim fac As Double, r As Double, V1 As Double, V2 As Double 
5       x = gauss * DesvPadX + XMedio 
        If LimiteDireito < x Or x < LimiteEsquerdo Then GoTo 5 
        Truncate_Normal_CV = x 
End Function 
 
'********************************************************************************** 
'*  Retorna números aleatórios de uma Distribuição Normal Truncada para Preço     * 
'********************************************************************************** 
Function Truncate_Normal_P(XMedio, DesvPadX, LimiteEsquerdo) 
    Dim x As Double 
    Dim fac As Double, r As Double, V1 As Double, V2 As Double 
5       x = gauss * DesvPadX + XMedio 
        If x < LimiteEsquerdo Then GoTo 5 
        Truncate_Normal_P = x 
End Function 
 
'********************************************************************************** 
'*          Retorna números aleatórios de uma Distribuição Normal Padrão          * 
'********************************************************************************** 
Function gauss() 
    Dim fac As Double, r As Double, V1 As Double, V2 As Double 
10      V1 = 2 * Rnd - 1 
        V2 = 2 * Rnd - 1 
        r = V1 ^ 2 + V2 ^ 2 
        If (r >= 1) Then GoTo 10 
        fac = Sqr(-2 * Log(r) / r) 



        gauss = V2 * fac 
End Function 
 
'*********************************************************************************** 
'*                           Sorteia os números                                    * 
'*********************************************************************************** 
Sub Sort(n As Variant, arr() As Double) 
    Dim Temp As Double 
    Dim i As Long 
    Dim j As Long 
    For j = 2 To n 
        Temp = arr(j) 
        For i = j - 1 To 1 Step -1 
            If (arr(i) <= Temp) Then GoTo 10 
            arr(i + 1) = arr(i) 
        Next i 
        i = 0 
10      arr(i + 1) = Temp 
    Next j 
End Sub 
  
'********************************************************************************** 
'*                    Constrói o Histograma da Distribuição                       * 
'********************************************************************************** 
Sub Hist(n As Variant, M As Long, Inicio As Double, Direito As Double, arr() As Double) 
    Dim i As Long, j As Long, Encontrar As Long 
    Dim Comprimento As Double 
    ReDim interrupcao(M) As Single 
    ReDim freq(M) As Single 
     
    For i = 1 To M 
        freq(i) = 0 
    Next i 
 
    Comprimento = (Direito - Inicio) / M 
     
    For i = 1 To M 
        interrupcao(i) = Inicio + Comprimento * i 
    Next i 
     
    For i = 1 To n 
        If (arr(i) <= interrupcao(1)) Then freq(1) = freq(1) + 1 
        If (arr(i) >= interrupcao(M - 1)) Then freq(M) = freq(M) + 1 
        For j = 2 To M - 1 
            If (arr(i) > interrupcao(j - 1) And arr(i) <= interrupcao(j)) Then freq(j) 
= freq(j) + 1 
        Next j 
    Next i 
     
    For i = 1 To M 
        Cells(i + 1, 9) = interrupcao(i) 
        Cells(i + 1, 10) = freq(i) 
    Next i 
End Sub 


